

bitest — Binomial probability test

Description	Quick start	Menu	Syntax
Option	Remarks and examples	Stored results	Methods and formulas
Reference	Also see		

Description

`bitest` performs exact hypothesis tests for binomial random variables. The null hypothesis is that the probability of a success on a trial is p . The total number of trials is the number of nonmissing values of *varname* (in `bitest`) or N (in `bitesti`). The number of observed successes is the number of 1s in *varname* (in `bitest`) or $\#_{\text{succ}}$ (in `bitesti`). *varname* must contain only 0s, 1s, and missing.

`bitesti` is the immediate form of `bitest`; see [\[U\] 19 Immediate commands](#) for a general introduction to immediate commands.

Quick start

Exact test for probability of success ($a = 1$) is 0.4

```
bitest a = .4
```

With additional exact probabilities

```
bitest a = .4, detail
```

Exact test that the probability of success is 0.46, given 22 successes in 74 trials

```
bitesti 74 22 .46
```

Menu

bitest

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > Binomial probability test

bitesti

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > Binomial probability test calculator

Syntax

Binomial probability test

```
bitest varname == #p [if] [in] [weight] [, detail]
```

Immediate form of binomial probability test

```
bitesti #N #succ #p [, detail]
```

by and collect are allowed with bitest; see [U] 11.1.10 **Prefix commands**.

fweights are allowed with bitest; see [U] 11.1.6 **weight**.

Option

Advanced

detail shows the probability of the observed number of successes, k_{obs} ; the probability of the number of successes on the opposite tail of the distribution that is used to compute the two-sided p -value, k_{opp} ; and the probability of the point next to k_{opp} . This information can be safely ignored. See the [technical note](#) below for details.

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

bitest
bitesti

bitest

► Example 1

We test 15 university students for high levels of one measure of visual quickness which, from other evidence, we believe is present in 30% of the nonuniversity population. Included in our data is *quick*, taking on the values 1 (“success”) or 0 (“failure”) depending on the outcome of the test.

```
. use https://www.stata-press.com/data/r18/quick
```

```
. bitest quick == 0.3
```

```
Binomial probability test
```

Variable	N	Observed k	Expected k	Assumed p	Observed p
quick	15	7	4.5	0.30000	0.46667
Pr(k >= 7)	= 0.131143	(one-sided test)			
Pr(k <= 7)	= 0.949987	(one-sided test)			
Pr(k <= 1 or k >= 7)	= 0.166410	(two-sided test)			

The first part of the output reveals that, assuming a true probability of success of 0.3, the expected number of successes is 4.5 and that we observed seven. Said differently, the assumed frequency under the null hypothesis H_0 is 0.3, and the observed frequency is 0.47.

The first line under the table is a one-sided test; it is the probability of observing seven or more successes conditional on $p = 0.3$. It is a test of $H_0: p = 0.3$ versus the alternative hypothesis $H_A: p > 0.3$. Said in English, the alternative hypothesis is that more than 30% of university students score at high levels on this test of visual quickness. The p -value for this hypothesis test is 0.13.

The second line under the table is a one-sided test of H_0 versus the opposite alternative hypothesis $H_A: p < 0.3$.

The third line is the two-sided test. It is a test of H_0 versus the alternative hypothesis $H_A: p \neq 0.3$. ◀

□ Technical note

The p -value of a hypothesis test is the probability (calculated assuming H_0 is true) of observing any outcome as extreme or more extreme than the observed outcome, with extreme meaning in the direction of the alternative hypothesis. In example 1, the outcomes $k = 8, 9, \dots, 15$ are clearly “more extreme” than the observed outcome $k_{\text{obs}} = 7$ when considering the alternative hypothesis $H_A: p \neq 0.3$. However, outcomes with only a few successes are also in the direction of this alternative hypothesis. For two-sided hypotheses, outcomes with k successes are considered “as extreme or more extreme” than the observed outcome k_{obs} if $\Pr(k) \leq \Pr(k_{\text{obs}})$. Here $\Pr(k = 0)$ and $\Pr(k = 1)$ are both less than $\Pr(k = 7)$, so they are included in the two-sided p -value.

The `detail` option allows you to see the probability (assuming that H_0 is true) of the observed successes ($k = 7$) and the probability of the boundary point ($k = 1$) of the opposite tail used for the two-sided p -value.

```
. bitest quick == 0.3, detail
```

```
Binomial probability test
```

Variable	N	Observed k	Expected k	Assumed p	Observed p
quick	15	7	4.5	0.30000	0.46667
Pr(k >= 7)	= 0.131143	(one-sided test)			
Pr(k <= 7)	= 0.949987	(one-sided test)			
Pr(k <= 1 or k >= 7)	= 0.166410	(two-sided test)			
Pr(k == 7)	= 0.081130	(observed)			
Pr(k == 2)	= 0.091560				
Pr(k == 1)	= 0.030520	(opposite extreme)			

Also shown is the probability of the point next to the boundary point. This probability, namely, $\Pr(k = 2) = 0.092$, is certainly close to the probability of the observed outcome $\Pr(k = 7) = 0.081$, so some people might argue that $k = 2$ should be included in the two-sided p -value. Statisticians (at least some we know) would reply that the p -value is a precisely defined concept and that this is an arbitrary “fuzzification” of its definition. When you compute exact p -values according to the precise definition of a p -value, your type I error is never more than what you say it is—so no one can criticize you for being anticonservative. Including the point $k = 2$ is being overly conservative because it makes the p -value larger yet. But it is your choice; being overly conservative, at least in statistics, is always safe. Know that `bitest` and `bitesti` always keep to the precise definition of a p -value, so if you wish to include this extra point, you must do so by hand or by using the `r()` stored results; see [Stored results](#) below. □

bitesti

▷ Example 2

The binomial test is a function of two statistics and one parameter: N , the number of observations; k_{obs} , the number of observed successes; and p , the assumed probability of a success on a trial. For instance, in a city of $N = 2,500,000$, we observe $k_{\text{obs}} = 36$ cases of a particular disease when the population rate for the disease is $p = 0.00001$.

```
. bitesti 2500000 36 .00001
```

```
Binomial probability test
```

N	Observed k	Expected k	Assumed p	Observed p
2,500,000	36	25	0.00001	0.00001
Pr(k >= 36)		= 0.022458 (one-sided test)		
Pr(k <= 36)		= 0.985448 (one-sided test)		
Pr(k <= 14 or k >= 36)		= 0.034859 (two-sided test)		

◀

Stored results

`bitest` and `bitesti` store the following in `r()`:

Scalars

<code>r(N)</code>	number N of trials
<code>r(P_p)</code>	assumed probability p of success
<code>r(k)</code>	observed number k of successes
<code>r(p_l)</code>	lower one-sided p -value
<code>r(p_u)</code>	upper one-sided p -value
<code>r(p)</code>	two-sided p -value
<code>r(k_opp)</code>	opposite extreme k
<code>r(P_k)</code>	probability of observed k (detail only)
<code>r(P_oppk)</code>	probability of opposite extreme k (detail only)
<code>r(k_nopp)</code>	k next to opposite extreme (detail only)
<code>r(P_noppk)</code>	probability of k next to opposite extreme (detail only)

Methods and formulas

Let N , k_{obs} , and p be, respectively, the number of observations, the observed number of successes, and the assumed probability of success on a trial. The expected number of successes is Np , and the observed probability of success on a trial is k_{obs}/N .

`bitest` and `bitesti` compute exact p -values based on the binomial distribution. The upper one-sided p -value is

$$\Pr(k \geq k_{\text{obs}}) = \sum_{m=k_{\text{obs}}}^N \binom{N}{m} p^m (1-p)^{N-m}$$

The lower one-sided p -value is

$$\Pr(k \leq k_{\text{obs}}) = \sum_{m=0}^{k_{\text{obs}}} \binom{N}{m} p^m (1-p)^{N-m}$$

If $k_{\text{obs}} \geq Np$, the two-sided p -value is

$$\Pr(k \leq k_{\text{opp}} \text{ or } k \geq k_{\text{obs}})$$

where k_{opp} is the largest number $\leq Np$ such that $\Pr(k = k_{\text{opp}}) \leq \Pr(k = k_{\text{obs}})$. If $k_{\text{obs}} < Np$, the two-sided p -value is

$$\Pr(k \leq k_{\text{obs}} \text{ or } k \geq k_{\text{opp}})$$

where k_{opp} is the smallest number $\geq Np$ such that $\Pr(k = k_{\text{opp}}) \leq \Pr(k = k_{\text{obs}})$.

Reference

Hoel, P. G. 1984. *Introduction to Mathematical Statistics*. 5th ed. New York: Wiley.

Also see

[R] [ci](#) — Confidence intervals for means, proportions, and variances

[R] [prtest](#) — Tests of proportions

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).